

BUILDING PRACTICE NOTE

PROGRAM FOR HAND-HELD COMPUTER
ESTIMATING ENERGY SAVINGS FROM REINSULATING HOUSES

by

R.L. Quirouette and E.C. Scheuneman

Division of Building Research, National Research Council of Canada

Ottawa, July 1982



PROGRAM FOR HAND-HELD COMPUTER
ESTIMATING ENERGY SAVINGS FROM REINSULATING HOUSES

by

R.L. Quirouette and E.C. Scheuneman

INTRODUCTION

Hand-held computers are readily available, at prices which make them attractive as computational aids. This program enables a hand-held computer to perform the energy audit calculations used in Building Practice Note No. 20, "Estimating Energy Savings From Reinsulating Houses"¹. The instruction package is designed for someone who knows little about computers but wishes to save time and effort on manual calculations. In this program, the computer prompts the user for information: it provides a fast and versatile way of performing energy audit calculations. The program uses the same input data and produces the same output results as in the Energy Savings Chart from Reference 1 (Appendix A). A person can easily change from a calculator to a hand-held computer. The symbols and terminology are the same.

Equipment

The program is written for the Sharp PC-1211 pocket computer. Other hand-held computers such as the Radio Shack TRS-80 can also be used. For the Hewlett Packard and Texas Instrument types of hand-held computers, however, the program would need to be rewritten and adapted.

The Computer Program

The program is organized into three general parts: the first calls for input data on the main building features, furnace characteristics, climate information and heat energy consumption records. The second part calls for a complete description of the building envelope components by area and thermal resistance, and the third part activates the analysis process to determine savings and cost benefits for various retrofit options.

The calculations performed by this program are in metric units. It is easily converted to Imperial units by changing one line of the program listing. The program should be keyed in as it is, however, so that it can be checked (debugged) against the sample, which is all metric.

Preparations for Loading the Program

Before proceeding with "loading" of the program, read the following notes carefully. The instructions must be followed precisely, as even a misplaced comma will stop ("hang up") the program.

Note the difference between 1 (the number one) and I (the letter I) in the program listing. The symbol 0 in the listing means the number

zero on the computer keyboard, whereas the symbol ϕ in the program listing means the letter O on the computer keyboard. Each program line begins with its Line Number and is followed by instructions that must be keyed in exactly as shown, with the same symbols and blank spaces. If an error is made while keying in a program line, press the clear (CL) key, rekey the entire line correctly and press ENTER. Or, by using the "◀", "▶", keys, move the blinking cursor over the letter or symbol to be changed, key the correct symbol over the error, and then move the cursor back to the end of the line. After ENTERING each line, the display will show the line in a slightly different form because the line input has been abbreviated to save computer storage space. For an explanation of the abbreviations, consult your computer manual.

TO LOAD THE PROGRAM INTO THE COMPUTER

- (1) Press the ON key.
- (2) Press the MODE key repeatedly until PRO appears on the display.
- (3) Key in the word NEW and press the ENTER key.
- (4) Key in the first Line of "the Program Listing" below, exactly as shown and press ENTER. After a few seconds, the Line will appear on the display in the non-abbreviated computer form.
- (5) Key in each successive Line and press ENTER until all the Lines are loaded into the computer.
- (6) Note, if you want to convert your program to Imperial units, repeat Line (1) and (2) and then key in the following Line:

2k=.0183:H=24:E=3413:G=1000:Y=167000

- (7) Proceed systematically to the end of the listing.

PROGRAM LISTING

```
1P."AUDIT"
2K=.361:H=86400:E=3600000:G=37300000:Y=38757000
3I."ADD=";D,"FB=";F,"φ/G/E=";Q$:IF Q$="φ"G.6
4IF Q$="E"LETY=E:G.6
5IF Q$="G"LETY=G
6I."K1=";U,"K3=";L,"VφL=";V,"A/C=";C
7"A":U."#####"
8I."# CMPS=";T:I=0:W=0:X=1
9P."CMP #=";X
10I."A=";A,"R1=";B
11P=A/B+.5:W=W+P-.5:A(30+X)=A:A(40+X)=B:I=I+1:X=X+1:P."Q1=";P:IF I<TG.9
12J=V*C*K+.5:P."Q11=";J:J=J+W:P."Q TφT=";J:J=J-.5
13"B":T=0:G.19
14I."CMP #=";X:U."#####":IF X<11G.17
15I."A/C=";C:A(70)=V*C*K+.5:P."Q11=";A(70):I."C=";A(28):φ=A(28)
16GφS.80:GφS.60:GφS.50:GφS.40:G.19
17I."R2=";R:P=A(30+X)/R+.5:P."Q2=";P:I."C=";A(28):A(50+X)=R:A(60+X)=A(28)
18GφS.80:GφS.70:GφS.50:GφS.40
19I."I SAV=";N$:IF N$="Y"G.14
20I."C SAV=";M$:IF M$="N"G.27
21I."# CMPS=";T:I=0:GφS.80
22I."CMP #=";X:I=I+1:IF X<11G.25
23A(28)=0:GφS.60:GφS.50:IF I<TG.22
24GφS.40:G.20
25GφS.70:GφS.50:IF I<TG.22
26GφS.40:G.20
27G.1
40U."#####":P."S1=";Q;" S2=";Z
41P."SP=";S;" C/S=";U."###.#";A(29):RE.
50Q=Q+(F*A/J*(1/B-1/R)):Z=Z+(H*D*U/(Y*L))*A*(1/B-1/R)):IF T>IG.54
51IF Z>=FLET S=(F-Q)*.75+Q:G.53
52S=Q+(Z-Q)*.75
53G=G+A(28):A(29)=G/S:G.55
54G=G+A(28)
55RE.
60A=1:B=1/(J-W):R=1/(V*C*K):RE.
70A=A(30+X):B=A(40+X):R=A(50+X):A(28)=A(60+X):RE.
80Q=0:Z=0:G=0:RE.
```

Testing and Running the Program

The computer program input should be tested against the sample calculations that follow to check that it is properly entered. For a detailed explanation of the display variables, see Appendix B which explains the prompt questions and the type of answers required.

If you make an error while keying in data for a "prompt" question and have not yet pressed the ENTER key, press the red (CL) key to clear your incorrect data from the display and then rekey the correct data. If you ENTER the wrong data you must restart that section (see User Notes & Tips). You will also need to put in a cost figure for the "C=" prompt question so that the computer can continue its calculation to get the

savings. If you don't know the actual cost, put in an approximate number. The only result affected by this number is the payback C/S.

The sample procedure that follows is taken from the Energy Savings Chart in Appendix A. It should be followed through step-by-step. If your results and the sample results are identical, the computer program is properly entered and it is ready to use with other building data.

- (1) Press ON.
- (2) Press the MODE key repeatedly until DEF appears on the display.
- (3) Key in the word CLEAR to remove all previous input data and press ENTER.
- (4) Key in the word RUN and press ENTER; the display will show AUDIT and the computer is now ready to accept the data from Appendix A.
- (5) Press ENTER to get ADD= on display.
- (6) Key in 5157 and press ENTER to get FB= on display.
- (7) Key in 1239 and ENTER to get O/G/E= on display.
- (8) Key in the letter ϕ and ENTER to get K1= on display.
- (9) Key in .218 and ENTER to get K3= on display.
- (10) Key in .55 and ENTER to get VOL= on display.
- (11) Key in 481.4 and ENTER to get A/C= on display.
- (12) Key in 1.00 and ENTER to get # CMPS= on display.
- (13) Key in 8 and ENTER to get CMP #= 1 on display. Press ENTER again to get A= on display.
- (14) Key in 65.0 and ENTER to get R1= on display.
- (15) Key in 2.11 and ENTER to get Q1= 31 on display. This number should coincide with the quantity in row 1, column Q1. Press ENTER to get CMP #= 2 on display. Press ENTER again to get A= on display.
- (16) Key in 143.1 and ENTER to get R1= on display.
- (17) Key in 1.76 and ENTER to get Q1= 81 on display. Press ENTER again to get CMP#= 3 on display. Refer to Appendix A for other values of A and R1 and continue the process until all 8 components are entered and Q11= 174 appears on display.
- (18) Press ENTER to get Q TOT= 506 on display.
- (19) Press ENTER to get I SAV= on display.
- (20) Key in Y and ENTER to get CMP #= on display.
- (21) Key in 1 and ENTER to get R2= on display.
- (22) Key in 5.64 and ENTER to get Q2= 12 on display. Press ENTER to get C= on display.
- (23) Key in 280 and press ENTER. After several seconds, you will see S1= 47 S2= 87 on display. Press ENTER to get SP= 77 C/S= 3.6 on display. Press ENTER to get I SAV= on display.
- (24) Repeat steps (20), (21), (22), and (23) but for component numbers (CMP#=) 2, 4, and 5 with the R2 and C data from each corresponding row.
- (25) Following the last entry for row 5, press ENTER to get I SAV= on display and key in Y and ENTER to get CMP #= on display.
- (26) Key in 11 and ENTER to get A/C= on display.
- (27) Key in .75 and ENTER to get Q11= 130 on display. Press ENTER to get C= on display.

- (28) Key in 200 and ENTER to get S1= 106 S2= 197 on display. Press ENTER to get SP= 175 C/S= 1.1 on display. Press ENTER to get I SAV= on display.
- (29) Key in N and ENTER to get C SAV= on display.
- (30) Key in Y and ENTER to get # CMPS= on display.
- (31) Key in 5 and ENTER to get CMP #= on display.
- (32) Key in 1 and ENTER to get CMP #= on display. After a few seconds CMP#= will reappear on display. Key in the remaining 4 component numbers 2, 4, 5, and 11 to get a total (row 12) S1= 476 S2= 887 on display. Press ENTER to get SP= 784 C/S= 3.1 on display. Press ENTER to get C SAV= on display.
- (33) Key in N and ENTER to get AUDIT on display which returns you to the beginning of the program.

Note, the data from the test example is stored in the computer for further use if desired (see User Notes & Tips).

SUMMARY

If all input data and output results have matched the numbers on the chart of Appendix A, the program is properly entered and is ready for use with other building data. If the program results are not identical with the chart numbers, run the example again in case you have inadvertently input some wrong data. However, if the input data was correct, then check the program listing on a line by line basis. The lines appearing on the display are not exactly the same as the program listing. This is the long form and you will have to consult your computer handbook for an explanation of the abbreviations and their meaning. The most common type of error will be a colon instead of a semicolon, a missing bracket or too many brackets, etc. If an error is found, rekey the whole line as in the program listing and try again.

User Notes & Tips

The real power of the computer becomes apparent when you want to study the effect of changing certain variables while leaving all other data the same. For example, you may want to consider several retrofit options, such as levels of insulation for a single component like the ceiling. This is easily done during the main procedure by specifying the same component number repeatedly when at the I SAV= part. It may also be done afterwards, by re-entering the program at the desired location. To start the program at the I SAV= routine only, press the SHFT key followed by the letter B, and when the display shows I SAV=, proceed as before. However, this will work only if there is building data stored in the computer. It is also possible to change the building component data while leaving all other data the same. Simply press the SHFT key, followed by the letter A and ENTER. This will restart the program at the building description part with # CMPS = on the display. Proceed as before.

To change the value of the other input data, such as FB (fuel bill), ADD (actual degree days), etc, while leaving the rest of the building data unchanged, use the appropriate procedure as defined in Appendix C.

REFERENCE

Quirouette, R.L. and Scheuneman, E.C. Estimating Energy Savings from Reinsulating Houses, National Research Council of Canada, Division of Building Research, Building Practice Note No. 20, 1980.

APPENDIX A

ENERGY SAVINGS CHART

(metric SI units)

NAME _____ TYPE OF HOUSE 2 STOREY, DETACHED INTERIOR TEMP. (T_i) 21°C
 ADDRESS _____ VOLUME OF HEATED SPACE 481.4 DEGREE-DAYS (DD) 4674
 CITY OTTAWA PROV. ONT. HEATING SYSTEM oil, FORCED AIR AIR CHANGE/h (No.) 1.0
 HOT WATER HEATING ELECTRIC
 FUEL BILL, $F_B =$ \$ 1239⁰⁰ $K_2 =$ 38 757 000
 $K_1 =$ \$ 0.218 / l $K_3 =$ 0.55

NO.	BUILDING COMPONENT	PREVIOUS YEAR			RETROFIT						
		A	R_1	$Q_1 = \frac{A}{R_1}$	R_2	$Q_2 = \frac{A}{R_2}$	S_1	S_2	S_p	C	C/S
1	CEILING	65.0	2.11	31	5.64	12	47	87	77	280 ⁰⁰	3.6
2	FRAME WALLS	143.1	1.76	81	2.29	62	46	85	75	1100 ⁰⁰	14.5
3	HEADER JOISTS	14.8	0.60	25							
4	BASEMENT WALLS, A.G.	19.5	0.26	75	2.47	8	164	305	270	450 ⁰⁰	1.6
5	BASEMENT WALLS, B.G.	49.2	0.8	62	3.2	15	112	210	185	450 ⁰⁰	2.4
6	BASEMENT FLOOR	65.0	4.6	14							
7	WINDOWS	13.9	0.35	40							
8	DOORS	3.7	0.7	5							
9											
10											
11	AIR CHANGE, $Q = 0.361 \times \text{Vol.} \times \text{No.}$		No. 1.0	174	No. .75	130	106	197	175	200 ⁰⁰	1.1
12	TOTAL			506			476	887	784	2480 ⁰⁰	3.1

$$S_1 = F_B \times \frac{(Q_1 - Q_2)}{\text{TOTAL } Q_1}$$

$$\text{ADD} = \text{DD} + (T_i - 18) \times b = 4674 + (21 - 18) \times 161 = 5157$$

$$S_2 = 86,400 \times \text{ADD} \times \frac{K_1}{K_2 \times K_3} \times (Q_1 - Q_2)$$

$$= 4.557 \times (Q_1 - Q_2)$$

(a) $\text{TOTAL } S_2 \leq F_B$

$$S_p = S_1 + 0.75 (S_2 - S_1)$$

(b) $\text{TOTAL } S_2 > F_B$

$$\text{TOTAL } S_p = \text{TOTAL } S_1 + 0.75 (F_B - \text{TOTAL } S_1)$$

APPENDIX B

<u>Prompt Question or output on display</u>	<u>Meaning</u>	<u>Reply</u>
AUDIT	program title	press ENTER
ADD=	actual degree days	usually a 4-digit number and ENTER
FB=	fuel bill in \$	usually a 3 or 4-digit number and ENTER
O/G/E=	oil, gas, or electric heating system	<u>φ</u> or <u>G</u> or <u>E</u> and ENTER
K1=	price in \$ per fuel unit	usually a decimal number and ENTER
K3=	seasonal efficiency of heating system	usually a decimal number and ENTER
VOL=	volume of heated space	usually a 3 or 4-digit number and ENTER
A/C	air change/hour	usually a number between 0 and 4 and ENTER
# CMPS=	number of building components excluding air change component #11	appropriate number from CHART and ENTER
CMP #- 1 (2,3,etc.)	data being requested for this component number	press ENTER
A=	requesting area for component being considered	appropriate number from CHART and ENTER
R1=	requesting R-value for component being considered	appropriate number from CHART and ENTER

Q1= 31	giving Q1 for component being considered	appears on CHART press ENTER (finish rest of components)
Q11= 174	giving Q1 for the air change rate	press ENTER
Q TOT= 506	giving the total Q for all components including an air change rate	press ENTER
I SAV=	do you want to calculate <u>I</u> ndividual <u>S</u> avings for a retrofit action	Y for yes or N for no and ENTER
CMP #=	what is the number of the component that has been retrofitted	appropriate number and ENTER (<u>Note</u> : to calculate for an Air Change retrofit, key in 11 and ENTER)
R2= or (AC/=)	what is the upgraded or new R-value of the component	appropriate number from CHART and ENTER
Q2= 12	giving Q2 for the upgraded component	press ENTER
C=	what is the cost of upgrading the component in \$	appropriate number from CHART and ENTER
S1= 47 S2= 87	giving the S1 and S2 savings in \$	press ENTER
SP= 77 C/S= 3.6	giving the SP savings in \$ and the payback in years	press ENTER

I SAV=	do you want to calculate <u>I</u> ndividual <u>S</u> avings for a retrofit action	key in N for no and press ENTER (after you have finished Individual <u>S</u> avings)
C SAV=	do you wish to calculate <u>C</u> ombined (TOTAL) <u>S</u> avings for retrofit actions	Y for yes or N for no and ENTER
# CMPS=	how many components were retrofitted for the <u>C</u> ombined <u>S</u> avings	appropriate number of components (5) and ENTER
CMP #=	what are the component numbers to be considered (this prompt repeats until all components are entered)	key in numbers 1, 2, 4, 5, 11, one at a time and ENTER after each entry
S1= 476 S2= 887	giving the <u>C</u> ombined (TOTAL) S1 and S2 <u>S</u> avings in \$	press ENTER
SP= 784 C/S= 3.1	giving the <u>C</u> ombined (TOTAL) SP <u>S</u> avings in \$ and combined (TOTAL) payback in years	press ENTER
C SAV=	do you wish to calculate <u>C</u> ombined (TOTAL) <u>S</u> avings for retrofit actions	N for no and ENTER (after you have finished <u>C</u> ombined (TOTAL) <u>S</u> avings)
AUDIT	calculations are finished and the program returns to the beginning	OFF if you are finished or ENTER to begin more calculations from the beginning

APPENDIX C

ADD (actual degree days)

- (1) Press ON twice.
- (2) Key in D= new value and ENTER.
- (3) Press SHFT A or SHFT B.

FB (fuel bill)

- (1) Press ON twice.
- (2) Key in F= new value and ENTER.
- (3) Press SHFT A or SHFT B.
NOTE... If the fuel bill is changed and the quantity of energy used remains the same then the unit fuel price (K1) must also be changed.

O/G/E (oil/gas/electricity)

- (1) Press ON twice.
- (2) Key in RUN and ENTER
- (3) Answer all questions again until VOL is encountered and then press SHFT A or SHFT B.

K1 (unit fuel price)

- (1) Press the ON key twice.
- (2) Key in U= new value and press ENTER.
- (3) Press SHFT B to go back into the I SAV part of the program or
- (4) Press SHFT A if the building components are to be input again or changed.
NOTE... If the unit fuel price is changed and the quantity of energy used remains the same then the fuel bill (FB) must also be changed.

K3 (seasonal efficiency)

- (1) Press ON twice.
- (2) Key in L= new value and ENTER.
- (3) Press SHFT A or SHFT B.

VOL (volume of heated space)

- (1) Press ON twice.
- (2) Key in V= new value and ENTER.
- (3) Key in RUN12 and ENTER to get Q11= number on display. Write this number in row 11 of the Q1 column and press ENTER to get Q TOT= number on display. Write this number in the TOTAL row of the Q1 column and ENTER to get I SAV= on display. Proceed with the desired calculations.

A/C (original air change)

- (1) Press ON twice.
- (2) Key in C= new value and ENTER.
- (3) Key in RUN12 and ENTER to get the value of Q11= on display. Write this number in row 11 of the Q1 column and press ENTER to get Q TOT= on display. Write this number in the TOTAL row of the Q1 column and press ENTER to get I SAV= on display. Proceed with the desired calculations.